

LES MARCHANDS DE SABLE


Il ne faut pas toujours beaucoup d'octets pour concevoir un jeu amusant. Le programme Basic présenté ici en est une preuve. Dirigées depuis le clavier du PB-700, des guêpes survolent un champ de fleurs. Leur tâche consiste à répandre le sommeil, à la manière des marchands de sable.

■ Si les possibilités graphiques du PB-700 méritent d'être soulignées, il faudra tout de même faire preuve d'imagination pour voir des guêpes et des fleurs sur l'écran.

Les premières ont pour mission d'endormir les secondes. Elles y parviennent en lançant un petit jet de poudre, commandé depuis le clavier. En touchant la fleur, ce jet l'endort définitivement. Une simple pression sur une touche — autre que BREAK, CAPS, STOP ou SHIFT — suffit pour lâcher de la poudre. Mais un nouveau jet ne peut avoir lieu que si la poudre du précédent a touché le sol.

Dès qu'une fleur est touchée, la guêpe a rempli sa mission et le score s'affiche. Si elle rase le champ de fleurs, elle repart pour une nouvelle mission après avoir touché le sol. L'exercice est alors plus périlleux que le précédent, jusqu'à ce que disparition s'ensuive.

La guêpe qui disparaît à droite de l'écran reviendra à gauche au tour suivant. Et, sous l'effet du sable, de nouvelles fleurs plongeront dans un lourd sommeil.



André REEB-GRUBER

Marchands de sable
 Programme pour PB-700
 Auteur André Reeb-Gruber
 Copyright LIST et l'auteur

```

*----- BLITZ -----*
5 CLEAR :CLS :M=8
10 DRAW(0,31)-(159,31):FOR I= 17 TO 137
STEP 6:Y=INT(RND*M)*2+2
15 FOR J=0 TO 4:DRAW(I+J,31)-(I+J,31-Y):
NEXT J:NEXT I:H=1:I=1
20 I=I+6:IF I>151 THEN I=7:H=H+2
25 IF H>28 THEN M=M+1:GOTO 10
30 DRAW(I-2,H-1)-(I-2,H)-(I+2,H)
35 IF POINT(I+4,H)=1 THEN 80
40 IF C<>0 THEN 65
45 IF INKEY$<>" " THEN D=5
50 GOTO 55+D
55 DRAW(I-2,H-1)-(I-2,H)-(I+2,H):GOTO 20
60 C=1:N=H:V=I
65 S=S+1:N=N+2:DRAW(V,N+2):DRAW(V-2,N)-
(V+2,N):DRAW(V-2,N+1)-(V+2,N+1):BEEP
70 IF N=29 THEN C=0:D=0:BEEP 1
75 GOTO 55
80 FOR Z=1 TO 100:NEXT Z:BEEP 1:BEEP 1:B
EEP 1:PRINT "PERDU:score ->";S
85 FOR I=1 TO 300:NEXT I
90 IF INKEY$=" " THEN 90 ELSE 5
  
```

Le programme ligne par ligne

- Ligne 5** : initialisation des mémoires, de l'écran et du niveau de difficulté.
Lignes 10 et 15 : tracé du sol et des fleurs, initialisation des coordonnées de la guêpe.
Ligne 20 : calcul des nouvelles coordonnées de la guêpe ; si elle est arrivée à la limite de l'écran, elle réapparaîtra au tour suivant avec une altitude plus faible.
Ligne 25 : si la guêpe a fini de raser le sol, le niveau de difficulté augmente d'une unité et l'on repart pour endormir de nouvelles fleurs.
Ligne 30 : tracé de la guêpe.
Ligne 35 : si la guêpe touche une fleur, envoi en ligne 80, c'est perdu.
Ligne 40 : test sur le jet de sable, grâce à un drapeau (C) ; si C est à zéro, il n'y a pas de jet de sable en cours ; si C est à 1, le sable est en train de tomber.
Ligne 45 : test de la commande de tir grâce au drapeau D ; si D vaut zéro, la commande de tir du jet n'est pas activée ; si D vaut 5, la commande est activée.
Ligne 50 : branchement à une adresse dépendant de la valeur de D.
Ligne 55 : effacement de la guêpe et retour à la ligne 20.
Ligne 60 : mise à 1 du drapeau C, initialisation des coordonnées de la guêpe qui permettent de repérer le déplacement du jet de sable.
Ligne 65 : incrémentation du score : tracé du sable qui endort éventuellement une fleur en émettant un « bip ».
Ligne 70 : si le sable a complètement endormi une fleur, les drapeaux C et D sont mis à zéro ; un nouveau « bip » retentit.
Lignes 80 : temporisation, signal sonore, affichage du résultat.

Liste des variables

- C** drapeau de « verrouillage » du jet de sable
D drapeau de commande du lancer de jet
H hauteur de la guêpe
I compteur de boucle (lignes 10 et 15) et position horizontale de la guêpe
J compteur de boucle
M niveau de difficulté
N hauteur du jet de sable
S score
V position horizontale du sable
Y hauteur de la fleur
Z compteur de boucle pour la temporisation.

